
An introduction to the GAMA platform

GAMA is a software platform for designing, programming, simulating, visualising 
and analysing spatially explicit agent-based models. It is free and open-source, 
fully documented, and comes bundled with ~200 example models and tutorials. 

Completely generic, but supporting specialized, 
domain specific plugins 

Own modeling language (GAML) and integrated 
development environment 

Powerful integration of spatial data (GIS), various 
agent architectures… 

Automated and/or guided exploration and 
calibration of models 

Developed under GPL license since 2007 by 
UMMISCO (IRD) and 6 French and Vietnamese 
institutions. 

Version 1.8 used by users worldwide, in various 
application fields 

http://gama-platform.org

An introduction to the use of the 
GAMA platform

It is now time to run GAMA !

First GAMA asks you to choose a 
workspace.  

A workspace is a folder that will 
contain all your own projects and 
models. 

You are free to choose the folder you 
want!

GAMA model files are stored in projects

Each project may contain several 
models, as well as additional 
resources (GIS data, pictures,…). 

Projects can be organised in any way, 
although a default layout is 
proposed: 

- includes : for all the ressource files 
- models : contains all the model files

GAMA model files are stored in projects

Each project may contain several 
models, as well as additional 
resources (GIS data, pictures,…). 

Projects can be organised in any way, 
although a default layout is 
proposed: 

- includes : for all the ressource files 
- models : contains all the model files

Problems view

Project Explorer

Editor view

Interactive 
console



Take a look at “Game of life” model in library

Open the model Life.gaml 
Library Models \ Toy models \ Life \ Life.gaml

The model has 
errors

The model can be 
experimented



Launching an experiment will switch from Modelling to 
Simula.on Perspective

Change the perspective

Simulation perspectiveModelling perspective

Exploring the Simulation perspective

Start/pause simulation (it will  
run until pause is clicked again)

Step the simulation (it will run one 
cycle of the simulation)

Adjust the speed of the simulation

Relaunch the simulation 
(necessary after having changed 
the parameter values)

Interrupt the simulation

Explore the simulation with parameters modified from 
Parameters view

The modifications made to the 
parameters are either: 

- Used for the current simulation when it 
makes sense (for instance, if the 
user changes a color) 

- Used when the user reloads the 
experiment otherwise (for instance, 
if the user changes the size of the 
grid) 

Launching experiment again (from 
the model editor) will erase the 
modifications. 

Inspect by right clicking on a agent in a display

Provides information about one specific agent.  
It also allows to change the values of its variables 

during the simulation.  

It is possible to «highlight» the selected agent.

Inspect informations by agent browser

The species browser provides informations about all or a 
selection of agents of a species.  

The agent browser is available through the Agents menu 
or by right clicking on a by right_clicking on a display

Take few minutes to explore some of the models of the 
Models Library.

Toy Models\ Segregation (Schelling) \ Segregation 
(GIS).gaml 

Experiment schelling

Library models \ Toy Models\ Ants (Foraging and 
Sorintg) \ Ant Foraging.gaml 

Experiment Classic

Library models \ Toy Models \ Boids \ Boids 3D 
Motion.gaml 

Experiment 3D

Library models \ Toy Models\ 
Traffic \ models  \ Simple 

Traffic Model.gaml 
Experiment trafic



0. Start developing a model: create a 
new project /import an existing one.

Import existing projects into the workspace

Imported projects are now  
in the User models



Creation of a new project

Creation of a new model file

model mymodel

global {
/** Insert the global definitions, variables and actions here */

}

experiment mymodel type: gui {
/** Insert here the definition of the input and output of the model */
output {
}

}

1

3

2



An introduction to the main concepts 
of the GAML language

Agent-based models are written using a few key 
concepts

Agents 

State  
(including a 
representation of 
the world and 
other agents). 

Behaviour 

Actions: 
Decision-making 
Communication 
Interactions 
Perception

AgentState Environment

Action DynamicsStateBehavior

Aspect

Agent-based classical meta-model

Environment 

State(s) 
Dynamics



Introduction to the main concepts of the  
GAMA Modelling Language - GAML

The role of GAML is to support modellers in writing models, which are 
specifications of simulations that can be executed and controlled 
during experiments, themselves specified by experiment plans. 

Agents in GAML are specified by their species, which provide them with 
a set of attributes (what they are, know…), actions (what they can do), 
behaviours (what they actually do) and also specifies properties of their 
population, for instance its topology 

Everything is an agent in GAML: the model itself (called the world), the 
agents defined in it, the experiments… 

Therefore, the structure of a model in GAML is simply a set of 
species declara.on statements 

3 types of block declaration 
(equivalent to species statements) 
are supported: 

- Global (unique): global attributes, 
actions, dynamics and initialisation. 

- Species and Grid: agent species. 
Several species statements can be 
defined in the same model. 

- Experiment : simulation execution 
context, in particular inputs and 
outputs. Several experiment 
statements can be defined.

model my_model

global {
/** Insert the global definitions, 
 * variables and actions here
 */

}

species my_species{
/** Insert here the definition of the 
 * species of agents
 */

}

experiment my_model type: gui {
/** Insert here the definition of the 
 * input and output of the model
 */

}

General Structure of a model Model

2 ways to write comments (texts that are not just part 
of the model but here for information purpose):

• //…  : for one line. Example : //this is a comment
• /* … */ : can be used for several lines. Example : /* 

this is as well a comment  */



Objective: 
 
Toward a bidirectional corridor model.

1. Creation of agents
handsOn 1.gaml

1. Creation of people agents

To do: We want to create and 
display 100 people agents. 

Steps to follow: 
- Definition of the people species 

- Creation of 100  people agents 
randomly located in the 
environment  

- Display of the agents

1. Creation of people agents 
Step 1: definition of the people species

To do: define the species people: 

Solution:

model my_model

global { }

species people{

}

experiment my_model type: gui { }

People species definition Model



1. Creation of people agents 
Step 2. Creation of 100 people agents

Hint: this is done at the initialization of the simulation, so in the init 
block of the global 

Solution:

model my_model

global {
init {

create people number: 100;
}

}

species people{ }

experiment my_model type: gui { }

Creation of 2000 people agents Global



1. Creation of people agents 
Step 3. Display of the people agents

Hint: the definition of the displays is made in an experiment 

Solution:

model my_model

global {
init {

create people number: 100;
}

}

species people{ }

experiment exp {
output {

display map {
species people;

}
}

}

Display the people agents Experiment



1. Creation of people agents 
Step 3. Display of the people agents

Result: people are only displayed as points, with a yellow color.

1. Creation of people agents  
Step 4. Define the way agents are displayed through an aspect

To do: define an aspect for the people agents 

Solution: define an aspect in the people species and use it in the 
display.

species people {
rgb col <- #blue;

aspect c {
draw circle(3) border: #black color: col;

}
}

Define an aspect for people agents People

experiment exp type: gui {
output {

display map {
species people aspect: c;

}
}

}

Display the people agents Experiment



1. Creation of people agents  
Step 4. Define the way agents are displayed through an aspect

To do: define an aspect for the people agents 

Solution: define an aspect in the people species and use it in the 
display.

species people {
rgb col <- #blue;

aspect c {
draw circle(3) border: #black color: col;

}
}

Define an aspect for people agents People

experiment exp type: gui {
output {

display map {
species people aspect: c;

}
}

}

Display the people agents Experiment

Addition of an attribute  
to the people species,  
containing the color  

used  to display the agent.



Creation of people agents  
Step 4. Define the way agents are displayed through an aspect

To do: define an aspect for the people agents 

Results:

The GAML corner: A statement represents either an 
imperative command or a declaration

Each line in a GAML model is a statement. 

It consists in a keyword, followed by a list of facets (some of them 
mandatory), ended by “;” or a block of statements. 

A facet is a keyword, followed by “:”, and an expression.  
- Note that the keyword of the first facet can usually be omitted.  

- If the statement is a declaration, the first facet contains an identifier. 

A block  is a set of statements enclosed into curly brackets (“{” and “}” )

33

global {
init {

create people number: 100;
}

}

species people{ }

Example of statements Model



The GAML corner: A statement represents either an 
imperative command or a declaration

Each line in a GAML model is a statement. 

It consists in a keyword, followed by a list of facets (some of them 
mandatory), ended by “;” or a block of statements. 

A facet is a keyword, followed by “:”, and an expression.  
- Note that the keyword of the first facet can usually be omitted.  

- If the statement is a declaration, the first facet contains an identifier. 

A block  is a set of statements enclosed into curly brackets (“{” and “}” )

global {
init {

create people number: 100;
}

}

species people{ }

Example of statements Model

keyword

Facet with  
a keyword (number)  

and  
an expression (2000)First Facet without  

a keyword



The GAML corner: A statement represents either an 
imperative command or a declaration

Each line in a GAML model is a statement. 

It consists in a keyword, followed by a list of facets (some of them 
mandatory), ended by “;” or a block of statements. 

A facet is a keyword, followed by “:”, and an expression.  
- Note that the keyword of the first facet can usually be omitted.  

- If the statement is a declaration, the first facet contains an identifier. 

A block  is a set of statements enclosed into curly brackets (“{” and “}” )

global {
init {

create people number: 100;
}

}

species people{ }

Example of statements Model

Examples of  
statement  
keywords keyword



The GAML Corner: example of statements

The GAML language 
contains many 
statements: 

- draw  

- create 

- loop 

- If - else 

- declaration 

- ask 

- … 

Example of the if - else:

if( condition ) {
set of statements to perform if the condition is true

} else {
set of statements to perform otherwise

}



The GAML corner: operators

Whereas statements are commands or 
declaration, operators are functions 
that compute a value on one or 
several operands. 

Unary operators are written: 
- operator(operand1) 

Binary operators are written: 
- Op1 operator Op2 

- operator(Op1, Op2) 

When there are more  
than 2 operands: 

- Op1 operator(Op2, …) 

- operator(Op1, Op2, …)

Operator circle computes   
a circle geometry  

with a given radius

species people {
rgb col <- #blue;

aspect c {
draw circle(3) border: #black color: col;

}
}

Define an aspect for people agents People



The GAML corner: built-in constants

GAML provides a set of built-in 
constants, starting with # 

- colors: #red, #yellow, #darkgrey… 

- units: #s, #h, #mn, # day, #m, #km… 

- mathematical: #pi, #e, #infinity… 

- Graphical units: #zoom, 
#camera_location

Creation of people agents  
Step 5. Addition of a parameter to control the initial number 
of people agents

global {
int init_nb_people <- 100;

init {
create people number: init_nb_people;

}
}

species people {
rgb col <- #blue;

aspect c {
draw circle(3) border: #black color: col;

}
}

experiment exp {

parameter "Nb people" var: init_nb_people min: 1 max: 10000;

output {
display map {

species people aspect: c;
}

}
}

Define an aspect for people agents Model
Addition of  

a global variable

Use of the global variable  
as parameter of the create

Make the global variable 
a parameter



All GAMA agents are provided with some 
built-in attributes :

• name (string)

• shape (geometry)

• location (point) : centroid of its shape 

The GAML Corner: definition of a species

4 kinds of elements can be defined 
in a species: 

- The internal state of the agents of this 
species (attributes). 

- Their capabilities (action): blocks that 
will be executed only when called. 

- Their behavior (reflex): blocks that will 
be executed at each step. 

- Their way of being displayed (aspect). 

In addition, an unique init block can 
be used to initialized agents at 
their creation. 

Note: global, grid, experiment are 
kinds of species and have the 
same structure. 40

species mon_espece{
string a_variable;

init { }
action my_action { }
reflex my_behavior { }
aspect my_aspect { }

}

General structure of a species Species



Notes on GAMA.

Every kind of agent has built-in attributes: 
name (a string) 

shape (a geometry)  
(default value = a point) 

location (a point) (value = the centroid of its shape)

global { }

species people {
rgb col ;

}



Space in GAMA

In GAMA, agents have a 
location in a reference 
continuous space. 

The reference continuous 
space is the shape of 
the world/global. 

It is defined by default as 
a 100m x 100m square. 

It can easily be redefined.

3000 m

3000m

global {
geometry shape <- square(3000#m);

}



Space in GAMA

Note: by default no 
grid (i.e. patches are 
created) 

But a grid can be 
added (that is a set of 
agents, with a square 
shape, whose cannot 
move …)

3000 m

3000m

global {
geometry shape <- square(3000#m);

}



Space in GAMA

In GAMA, agents have 
a location in a 
reference 
continuous space.

3000 m

3000m



Space in GAMA

In GAMA, agents have 
a location in a 
reference 
continuous space.

45

3000 m

3000m

name = ''people2'’
location = {1400.1, 44.02, 0.0} 
shape = {1400.1, 44.02, 0.0}



Summary of the model 1

46

global {
int init_nb_people <- 100;

init {
create people number: init_nb_people;

}
}

species people {
rgb col <- #blue;

aspect c {
draw circle(3) border: #black color: col;

}
}

experiment exp {

parameter "Nb people" var: init_nb_people min: 1 max: 10000;

output {
display map {

species people aspect: c;
}

}
}

Define an aspect for people agents Model

Try to run the simulation!



2. Random move
handsOn 2.gaml

2. Random move 
General remark

As any agent has a location, make an agent move is only changing 
its location. 

2 options alternatives are used below: (i) "jump" in a random location 
of the environment, (ii) a step in close neighbourhood. 

This requiers to define a behaviour to the agents, using a reflex.

species people {
rgb col <- #blue;

reflex move {
location <- any_location_in(world.shape);

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People
species people {

rgb col <- #blue;

reflex move {
location <- any_location_in(world.shape);

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People



2. Random move 
Alternatives of random move

species people {
rgb col <- #blue;

reflex move {
location <- { location.x + rnd(-0.5,0.5) , location.y + rnd(-0.5,0.5)};

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People

species people {
rgb col <- #blue;

reflex move {
location <- any_location_in(world.shape);

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People



species people {
rgb col <- #blue;

reflex move {
location <- any_location_in(world.shape);

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People

2. Random move 
Define a random move for agents (i)

To do: define a random move behavior for the people agents 

Hints: for an agent to move is simply to change its location.  

Solution:

Reflex is used  
to define a behavior

Compute a random  
location in a geometry

world is the global agent.  
Its shape is the spatial  
environment in which  
agents are located.



2. Random move 
Define a random move for agents (ii)

To do: define a random move behavior for the people agents 

Hints: for an agent to move is simply to change its location.  

Solution:

species people {
rgb col <- #blue;

reflex move {
location <- { location.x + rnd(-0.5,0.5) , location.y + rnd(-0.5,0.5)};

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People
Built-in attribute 

location is of type  
point.

A point is a  
complex type,  

with 3 component:  
x, y, z

A point can be defined by its literal definition: {a_x, a_y} or {a_x, a_y, a_z}.



2. Random move 
Define a random move for agents (ii)

To do: define a random move behavior for the people agents 

Hints: for an agent to move is simply to change its location.  

Solution:

species people {
rgb col <- #blue;

reflex move {
location <- { location.x + rnd(-0.5,0.5) , location.y + rnd(-0.5,0.5)};

}

aspect c {
draw circle(3) border: #black color: col;

}
}

Define a move behavior People

Run the simulation!



The GAML Corner:  
Species are provided with a simple behavioural structure, based on 
reflexes (what they actually do)

A reflex is a sequence of statements that can be executed, at each time 
step, by the agent.  

reflex name when: condition{
[statements]

}

If no facet when are defined, it will be executed every time step.  

If there is one, it is executed only if the boolean expression evaluates to 
true.  

Several reflex blocks can be defined in each species. Each will be 
executed at each simulation step.

Note: The init block is a specific 
reflex that is activated only once at 
the creation of the agent



3. Perception and move
handsOn 3.gaml

3. Perception and move 
Move under the influence of other agents

Objective: 
Introduce a separation force to prevent for people agent collisions in 
their random move. 

Steps: 
- Agents move will be directed by a "direction" vector, randomly chosen at every 

simulation step. 

- Agents observe the agents in  their neighbourhood (at a given distance) and 
compute a resulting "repulsion force". 

- Agents move results from the composition of these 2 vectors.

Direction

Repulsion

Repulsion

Result



3. Perception and move 
Move under the influence of other agents

56

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - ((location) - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior People



3. Perception and move 
Step 1. Direction vector

To do: Agents move are directed by a "direction" vector, randomly 
chosen at every simulation step.

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - (location - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior People

Random init of the  
direction vector,  

at each step.

Location change  
computed from  

the direction vector



3. Perception and move 
Step 2. Repulsion vector

To do: Agents observe the agents in  their neighbourhood (at a given 
distance) and compute a resulting "repulsion force".

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - (location - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior People
The agent perceives  
the people agents 
located at a given  
distance of itself  
(i.e. overlapping  
a circle around it)

global {
 int init_nb_people <- 100; 
 float minimal_distance <- 10.0;  
}

Global variable Global



3. Perception and move 
Step 2. Repulsion vector

To do: Agents observe the agents in  their neighbourhood (at a given 
distance) and compute a resulting "repulsion force".

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - (location - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior PeopleAsk agent_set {} 
is a loop over 

a set of agents. 
In the ask, the execution 

context is the one of  
each agent. 

 
I.e. location is the location  

of each agent of agent_set. 
The location of the  

calling agent is  
myself.location



3. Perception and move 
Step 2. Repulsion vector

To do: Agents observe the agents in  their neighbourhood (at a given 
distance) and compute a resulting "repulsion force".

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - (location - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior People
Definition of an action 

returning a value  
of type point

Call of the action.



The GAML corner: an action in GAML is a capability available 
to the agents of a species (what they can do)

It is a block of statements that can be used 
and reused whenever needed. An action 
can accept arguments.  

An action can return a result (statement 
return). 

return_type action_name (var_type arg_name,...) 
 {  
  [statements]  
  [return value;]  
}

Some actions are directly available (built-in 
action, i.e. primitive) for all agents (e.g. 
die action) or to specific agents (pause 
action of the world agents)

action simple_action {
  write "simple action";
}

int   sum (int a <- 100, int b) {
  return a + b;
}

Action that returns a value

write statement displays  
a message in the console

Type of the  
returned value

Return a value,  
and finish the action

Definition 2 arguments :  
the first one named «a», type integer  

and default value is «100»;  
the second named «b», type integer



The GAML Corner: Different ways to call an action in GAML

Call a action that does not return any value: 
 
do action_name(v1,v2);

Call an action that returns a value: 
 
my_var <- self.action_name(arg1:v1, arg2:v2); 

Examples: 

do action_simple;

int d <- self.add(10,100);

62



3. Perception and move 
Step 3. Move from resulting vectors

To do: Agents move results from the composition of these 2 vectors.

species people {
rgb col <- #blue;

point separation  {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - (location - myself.location);
}  
return rep;

}

reflex move {
point direction <- {rnd(-0.5,0.5),rnd(-0.5,0.5)};
direction <- direction + self.separation();
location <- location + direction/(3*norm(direction));

}

Define a move behavior People

Composition of  
the 2 vectors

Update location.



4. Move toward a target 
handsOn 4.gaml

4. Move toward a target

Objective: 
Introduce an additional "force" toward a given target. 

Steps: 
- Agents get an additional attribute: target, initialised on the border of the 

environment. 

- Agents are "attracted" by their target. 

- When agents reach their target, they are removed from the simulation.

4. Move toward a target 
Step 1. Add a target attribute and initialise it

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

species people {
rgb col <- #blue;
point target;

init {
if( flip(0.5)) {

target <- flip(0.5) ? {0, rnd(100)} : {100, rnd(100)};
} else {

target <- flip(0.5) ? {rnd(100) , 0} : {rnd(100), 100};
}

}

Define a move behavior People



4. Move toward a target 
Step 1. Add a target attribute and initialise it

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

67

species people {
rgb col <- #blue;
point target;

init {
if( flip(0.5)) {

target <- flip(0.5) ? {0, rnd(100)} : {100, rnd(100)};
} else {

target <- flip(0.5) ? {rnd(100) , 0} : {rnd(100), 100};
}

}

Define a move behavior People

The operator ? : 
Condition ? valueIfTrue : valueIfFalse

Operator flip, that computes   
randomly the value  

true or false  
with a given probability



The GAML corner: Init block 

For each species, an init block can 
be defined 

It allows to execute a sequence of 
statements at the creation of the 
agents 

Activated only once when the 
agent is created, after the 
initialisation of its variables, and 
before it executes any reflex 

Only one instance of init per species

global {
............
//Only executed when world agent is 

created
init { 
    write "Executing initialisation";

}
}



4. Move toward a target 
Step 2. Attraction by the target

To do: Agents are "attracted" by their target.

species people {
rgb col <- #blue;
point target;

point separation {
point rep <- {0,0};
ask (people overlapping (circle(minimal_distance)))  {

rep <- rep - ((location) - myself.location);
}  
if rep  != {0,0} {

return rep / norm(rep);
} else {

return rep;
}

}

point attraction_target {
return (target - location) / norm(target - location);

}

reflex move {
point direction <- self.attraction_target();
direction <- direction + self.separation() * 3 ;
location <- location + direction/(norm(direction));

}

Define a move behavior People



4. Move toward a target 
Step 3. Remove agents that reached their target

To do: When agents reach their target, they are removed from the 
simulation.

species people {
rgb col <- #blue;
point target;

reflex death {
if( location distance_to target < 1 #m ) {

do die;
}

}

}

Define a move behavior People



4. Move toward a target 
Step 3. Remove agents that reached their target

To do: When agents reach their target, they are removed from the 
simulation.

species people {
rgb col <- #blue;
point target;

reflex death {
if( location distance_to target < 1 #m ) {

do die;
}

}

}

Define a move behavior People

die is a built-in action of any agent. 
It destroys the agent.



5. Using built-in skills 
handsOn 5.gaml

72



5. Using built-in skills

Note: 
The force-based pedestrian behaviour is a behaviour that can be 
reused in many model. 

- As consequence, it has been integrated inside GAMA and can be used easily. 

GAML allows modeler to add  some  skills to each agent of a species. 
- This adds new possible actions and attributes  

E.g. GAMA provides: 
- a moving skill to let agents move on a grid, graph or continuous 

space toward a target, or wander 
- A driving skills to let agents moving on a road graph, with several 

lanes, signals… 
- A communicating skill to use network feature to communicate …. 
- and a pedestrian skill implementing the Social Force Model. 73



5. Using built-in skills

Objective: 
Replace the computation of  the various forces by the built-in 
pedestrian skill. 

Steps: 
- Add the skill to the people species. 

- Remove all the vector computation 

- Use the proper action to walk toward the target.

5. Using built-in skills 
Step 1. Add the skill to the people species.

To do: Assign the pedestrian skill to the people species.

75

Add pedestrian skill.

species people skills: [pedestrian]{
rgb col <- #blue;
point target;

 // … 
}

Add a skill People



5. Using built-in skills 
Step 2. Use the pedestrian skill 

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

species people skills: [pedestrian]{
rgb col <- #blue;
point target;

init {
if( flip(0.5)) {

target <- flip(0.5) ? {0, rnd(100)} : {100, rnd(100)};
} else {

target <- flip(0.5) ? {rnd(100) , 0} : {rnd(100), 100};
}

pedestrian_species <- [people];
pedestrian_model <- "advanced";
pedestrian_consideration_distance <- P_pedestrian_consideration_distance;
shoulder_length <- P_shoulder_length;

}

reflex move {
do walk_to target: target bounds:free_space;

}

// …
}

Define a move behavior People



5. Using built-in skills 
Step 2. Use the pedestrian skill 

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

species people skills: [pedestrian]{
rgb col <- #blue;
point target;

init {
if( flip(0.5)) {

target <- flip(0.5) ? {0, rnd(100)} : {100, rnd(100)};
} else {

target <- flip(0.5) ? {rnd(100) , 0} : {rnd(100), 100};
}

pedestrian_species <- [people];
pedestrian_model <- "advanced";
pedestrian_consideration_distance <- P_pedestrian_consideration_distance;
shoulder_length <- P_shoulder_length;

}

reflex move {
do walk_to target: target bounds:free_space;

}

// …
}

Define a move behavior People

Initialize  
parameters  
of the SFM

Call the walk_to  
action: 

it computes  
the various forces



5. Using built-in skills 
Step 2. Use the pedestrian skill 

To do: define the parameters of the SFM as global variables and 
parameters of the simulation.

78

global {
int init_nb_people <- 100;
float minimal_distance <- 10.0;

geometry free_space <- world.shape;

float P_shoulder_length <- 2.0 parameter: true;
float P_pedestrian_consideration_distance <- 7.0 parameter: true ;

init {
create people number: init_nb_people;

}
}

Additional global variables Global

The space in which  
agents can walk. 

 
It is set to the whole  

environment



6. Pedestrians in a corridor
model 2.gaml

6. Pedestrians in a corridor 

Objective: 
This model reproduces a bidirectional stream of pedestrian in a 
corridor. 

Remark: 
The agents are created a given simulation step and at a location 
specified in a csv file ( init_simstep0.0625_bi_corr_400_b_10.txt ) 

- The model can display trajectories (and also save them).

7. Data importation in GAMA

81



What are Geographic Informa7on Systems (GIS)?

- A computer system designed to capture, 
store, manipulate, analyse, manage, and 
present all types of geographical data.

82

The real  
world 

Database 



Representa7on of geographical data

83Vector Raster

Real world

Vector Raster

– Raster (grid or image): A geographical 
phenomena is represented as a par;;on 
(in cells) of the geographical space. Each 
cell has one or several a>ributes that 
define its content. 

– Vector: A geographical phenomena is 
represented by one or several geometric 
primi;ves (point, line, polygons), 
described by a list of coordinates and an 
interpola;on func;on. A classic vector 
format filed is the shapefile: it is 
composed of 4 main 
files: .shp, .dbf, .shx, .prj



A@ribute data – non spa7al data

- In raster datasets, informa;on associated 
with each unique value of a raster cell.  

- In vector dataset, data stored in a table 
and linked to each object by a unique 
iden;fier

84

Attribute of a raster format

Attribute of vector format



Every agent in GAMA has a geometry (its shape).

An agent’s geometry can be:  
- a point (default),  
- a polyline,  
- a polygon or  
- a complex geometry (2D-3D) 

An agent geometry is accessible through 
GAML thanks to the “shape” built-in 
a>ribute. 

World agent (<<global>> species) have also a 
“shape” built-in a>ribute which define the 
shape of environment 

Agents can be created directly from GIS shapefile

create my_species from: my_shapefile;

Each GIS object  
becomes an agent of  
the specified species

my_shapefile

my_species



The shape of the “world” agent represents the global 
environment of a model

By default, a model global environment is a 
2D square space of 100m x 100m 

It is also possible to redefine the global 
environment using a shapefile or asc file.

Environment

global{
file river_shapefile <- file("../includes/river.shp");
geometry shape <- envelope(river_shapefile);

}



8. Pedestrian model in data-based 
environment
Plugin Models/Pedestrian  skill/models/Complex 
environment - walk.gaml

8. Pedestrian model in data-based environment  
Main concepts

The pedestrian skill allows the agents to move  on a complex 
environment using a graph (paving the environment) to reach their 
target, avoiding  other agents (using SFM algorithm). 

GAMA provides tools to compute this graph from environment data  
(either the walls for inside move or buildings for outside move)

89



8. Pedestrian model in data-based environment  
Data

Walls



8. Pedestrian model in data-based environment  
Data

Walls

Open area



8. Pedestrian model in data-based environment  
Data

Walls Pedestrian  path

Open area



8. Pedestrian model in data-based environment  
Data

Walls Pedestrian  path

Open area Free space



8. Pedestrian model in data-based environment  
Step 2. Use the pedestrian skill 

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

species people skills: [pedestrian]{
rgb color <- rnd_color(255);
float speed <- gauss(5,1.5) #km/#h min: 2 #km/#h;

reflex move  {
if (final_waypoint = nil) {

do compute_virtual_path pedestrian_graph:network  
                                   target: any_location_in(open_area) ;

}
do walk ;

}
//  …
}

Define a move behavior People

When the targe is nil,  
compute  a new one,  

 and the path to reach it



8. Pedestrian model in data-based environment  
Step 2. Use the pedestrian skill 

To do: Agents get an additional attribute: target, initialised on the 
border of the environment.

species people skills: [pedestrian]{
rgb color <- rnd_color(255);
float speed <- gauss(5,1.5) #km/#h min: 2 #km/#h;

reflex move  {
if (final_waypoint = nil) {

do compute_virtual_path pedestrian_graph:network  
                                   target: any_location_in(open_area) ;

}
do walk ;

}
//  …
}

Define a move behavior People

walk will compute the next  location,  
given the network,  

other agents location,  
simulation step… 



9. Multi-level model
Library Models/Modeling/Multi-Level Usage/Corridor.gaml

97http://gama-platform.org

http://gama-platform.org

